# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=quickcheck-io
pkgname=haskell-quickcheck-io
pkgver=0.2.0
pkgrel=613
pkgdesc="Use HUnit assertions as QuickCheck properties"
url="https://github.com/hspec/quickcheck-io"
license=('MIT')
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-hunit' 'haskell-quickcheck')
makedepends=('ghc')
source=(https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz)
sha512sums=('02e1f40bac8f31f27c21d1d2df9f226816f9da871092b09f0451a8321e250c9e94887f94889be421c9d0034e0a27b02c907d16eb7da5a4376df42c7c1fbd01df')

build() {
    cd $_hkgname-$pkgver
    
    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

package() {
    cd $_hkgname-$pkgver
    
    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
